Instructions:
Language of the Computer

Euiseong Seo (euiseong@skku.edu)
Computer Systems Laboratory
Sungkyunkwan University
http://csl.skku.edu
Instruction Set

- The repertoire of instructions of a computer
- Different computers have different instruction sets
  - But with many aspects in common
- Early computers had very simple instruction sets
  - Simplified implementation
- Many modern computers also have simple instruction sets
The MIPS Instruction Set

- Used as the example throughout the book
- Stanford MIPS commercialized by MIPS Technologies (www.mips.com)
- Large share of embedded core market
  - Applications in consumer electronics, network/storage equipment, cameras, printers, …
- Typical of many modern ISAs
  - See MIPS Reference Data tear-out card, and Appendices B and E
Arithmetic Operations

- Add and subtract, three operands
  - Two sources and one destination
    \[ \text{add } a, b, c \quad \# \quad a \text{ gets } b + c \]
- All arithmetic operations have this form

**Design Principle 1:**
Simplicity favors regularity
  - Regularity makes implementation simpler
  - Simplicity enables higher performance at lower cost
Arithmetic Example

- C code:
  \[ f = (g + h) - (i + j); \]

- Compiled MIPS code:
  ```
  add t0, g, h       # temp t0 = g + h
  add t1, i, j       # temp t1 = i + j
  sub f, t0, t1      # f = t0 - t1
  ```
Register Operands

- Arithmetic instructions use register operands
- MIPS has a 32 × 32-bit register file
  - Use for frequently accessed data
  - Numbered 0 to 31
  - 32-bit data called a “word”
- Assembler names
  - $t0, $t1, …, $t9 for temporary values
  - $s0, $s1, …, $s7 for saved variables
- Design Principle 2: Smaller is faster
  - c.f. main memory: millions of locations
Register Operand Example

- C code:
  
  ```c
  f = (g + h) - (i + j);
  ```
  
  - f, ..., j in $s0, ..., $s4

- Compiled MIPS code:
  
  ```mips
  add $t0, $s1, $s2  
  add $t1, $s3, $s4  
  sub $s0, $t0, $t1  
  ```
Memory Operands

- Main memory used for composite data
  - Arrays, structures, dynamic data
- To apply arithmetic operations
  - Load values from memory into registers
  - Store result from register to memory
- Memory is byte addressed
  - Each address identifies an 8-bit byte
- Words are aligned in memory
  - Address must be a multiple of 4
- MIPS is Big Endian
  - Most-significant byte at least address of a word
  - c.f. Little Endian: least-significant byte at least address
Memory Operand Example 1

- **C code:**
  
  ```
  g = h + A[8];
  ```
  
  - g in $s1, h in $s2, base address of A in $s3

- **Compiled MIPS code:**
  
  - Index 8 requires offset of 32
    - 4 bytes per word
  
  ```
  lw  $t0, 32($s3)    # load word
  add $s1, $s2, $t0
  ```

offset \quad base register
Memory Operand Example 2

- **C code:**
  
  \[ A[12] = h + A[8]; \]
  
  - \( h \) in \$s2, base address of \( A \) in \$s3

- **Compiled MIPS code:**
  
  - Index 8 requires offset of 32
  
  ```
  lw $t0, 32($s3)    # load word
  add $t0, $s2, $t0
  sw $t0, 48($s3)    # store word
  ```
Registers vs. Memory

- Registers are faster to access than memory
- Operating on memory data requires loads and stores
  - More instructions to be executed
- Compiler must use registers for variables as much as possible
  - Only spill to memory for less frequently used variables
  - Register optimization is important!
Immediate Operands

- Constant data specified in an instruction
  \[
  \text{addi } s3, s3, 4
  \]
- No subtract immediate instruction
  - Just use a negative constant
    \[
    \text{addi } s2, s1, -1
    \]
- **Design Principle 3: Make the common case fast**
  - Small constants are common
  - Immediate operand avoids a load instruction
The Constant Zero

- MIPS register 0 ($zero) is the constant 0
  - Cannot be overwritten
- Useful for common operations
  - E.g., move between registers
    add $t2, $s1, $zero
### Unsigned Binary Integers

- **Given an n-bit number**
  \[ x = x_{n-1}2^{n-1} + x_{n-2}2^{n-2} + \cdots + x_12^1 + x_02^0 \]

- **Range:** 0 to \( 2^n - 1 \)

- **Example**
  - 0000 0000 0000 0000 0000 0000 0000 1011
    \[ = 0 + \ldots + 1 \times 2^3 + 0 \times 2^2 + 1 \times 2^1 + 1 \times 2^0 \]
    \[ = 0 + \ldots + 8 + 0 + 2 + 1 = 11_{10} \]

- **Using 32 bits**
  - 0 to \( +4,294,967,295 \)
2s-Complement Signed Integers

- Given an n-bit number

\[ x = -x_{n-1}2^{n-1} + x_{n-2}2^{n-2} + \ldots + x_12^1 + x_02^0 \]

- Range: \(-2^{n-1}\) to \(+2^{n-1} - 1\)

- Example

  1111 1111 1111 1111 1111 1111 1111 1100₂
  
  \[ = -1 \times 2^{31} + 1 \times 2^{30} + \ldots + 1 \times 2^2 + 0 \times 2^1 + 0 \times 2^0 \]
  
  \[ = -2,147,483,648 + 2,147,483,644 = -4_{10} \]

- Using 32 bits

  - \(-2,147,483,648\) to \(+2,147,483,647\)
2s-Complement Signed Integers

- Bit 31 is sign bit
  - 1 for negative numbers
  - 0 for non-negative numbers
- $-(2^{n-1})$ can’t be represented
- Non-negative numbers have the same unsigned and 2s-complement representation
- Some specific numbers
  - 0: 0000 0000 ... 0000
  - –1: 1111 1111 ... 1111
  - Most-negative: 1000 0000 ... 0000
  - Most-positive: 0111 1111 ... 1111
Signed Negation

- **Complement and add 1**
  - Complement means 1 → 0, 0 → 1

\[
\overline{x} + x = 1111...111_2 = -1
\]

\[
\overline{x} + 1 = -x
\]

- **Example: negate +2**
  - \(+2 = 0000 0000 ... 0010_2\)
  - \(-2 = 1111 1111 ... 1101_2 + 1 = 1111 1111 ... 1110_2\)
Sign Extension

- Representing a number using more bits
  - Preserve the numeric value
- In MIPS instruction set
  - addi: extend immediate value
  - lb, lh: extend loaded byte/halfword
  - beq, bne: extend the displacement
- Replicate the sign bit to the left
  - c.f. unsigned values: extend with 0s
- Examples: 8-bit to 16-bit
  - +2: 0000 0010 => 0000 0000 0000 0010
  - –2: 1111 1110 => 1111 1111 1111 1110
Representing Instructions

- Instructions are encoded in binary
  - Called machine code

- MIPS instructions
  - Encoded as 32-bit instruction words
  - Small number of formats encoding operation code (opcode), register numbers, …
  - Regularity!

- Register numbers
  - $t0 – t7$ are reg’s 8 – 15
  - $t8 – t9$ are reg’s 24 – 25
  - $s0 – s7$ are reg’s 16 – 23
MIPS R-format Instructions

- **Instruction fields**
  - **op**: operation code (opcode)
  - **rs**: first source register number
  - **rt**: second source register number
  - **rd**: destination register number
  - **shamt**: shift amount (00000 for now)
  - **funct**: function code (extends opcode)
R-format Example

<table>
<thead>
<tr>
<th>op</th>
<th>rs</th>
<th>rt</th>
<th>rd</th>
<th>shamt</th>
<th>funct</th>
</tr>
</thead>
<tbody>
<tr>
<td>6 bits</td>
<td>5 bits</td>
<td>5 bits</td>
<td>5 bits</td>
<td>5 bits</td>
<td>6 bits</td>
</tr>
</tbody>
</table>

add $t0, $s1, $s2

<table>
<thead>
<tr>
<th>special</th>
<th>$s1</th>
<th>$s2</th>
<th>$t0</th>
<th>0</th>
<th>add</th>
</tr>
</thead>
<tbody>
<tr>
<td>0</td>
<td>17</td>
<td>18</td>
<td>8</td>
<td>0</td>
<td>32</td>
</tr>
<tr>
<td>000000</td>
<td>10001</td>
<td>10010</td>
<td>01000</td>
<td>00000</td>
<td>100000</td>
</tr>
</tbody>
</table>

00000010001100100100000000100000_2 = 02324020_{16}
Hexadecimal

- **Base 16**
  - Compact representation of bit strings
  - 4 bits per hex digit

<table>
<thead>
<tr>
<th></th>
<th>0000</th>
<th>0100</th>
<th>1000</th>
<th>c</th>
<th>1100</th>
</tr>
</thead>
<tbody>
<tr>
<td>0</td>
<td>0000</td>
<td>0100</td>
<td>1000</td>
<td>c</td>
<td>1100</td>
</tr>
<tr>
<td>1</td>
<td>0001</td>
<td>0101</td>
<td>1001</td>
<td>d</td>
<td>1101</td>
</tr>
<tr>
<td>2</td>
<td>0010</td>
<td>0110</td>
<td>1010</td>
<td>e</td>
<td>1110</td>
</tr>
<tr>
<td>3</td>
<td>0011</td>
<td>0111</td>
<td>1011</td>
<td>f</td>
<td>1111</td>
</tr>
</tbody>
</table>

**Example: eca8 6420**
- 1110 1100 1010 1000 0110 0100 0010 0000
MIPS I-format Instructions

<table>
<thead>
<tr>
<th>op</th>
<th>rs</th>
<th>rt</th>
<th>constant or address</th>
</tr>
</thead>
<tbody>
<tr>
<td>6 bits</td>
<td>5 bits</td>
<td>5 bits</td>
<td>16 bits</td>
</tr>
</tbody>
</table>

- **Immediate arithmetic and load/store instructions**
  - rt: destination or source register number
  - Constant: $-2^{15}$ to $+2^{15} - 1$
  - Address: offset added to base address in rs

- **Design Principle 4: Good design demands good compromises**
  - Different formats complicate decoding, but allow 32-bit instructions uniformly
  - Keep formats as similar as possible
Stored Program Computers

- Instructions represented in binary, just like data
- Instructions and data stored in memory
- Programs can operate on programs
  - e.g., compilers, linkers, ...
- Binary compatibility allows compiled programs to work on different computers
  - Standardized ISAs
## Logical Operations

**Instructions for bitwise manipulation**

<table>
<thead>
<tr>
<th>Operation</th>
<th>C</th>
<th>Java</th>
<th>MIPS</th>
</tr>
</thead>
<tbody>
<tr>
<td>Shift left</td>
<td>&lt;&lt;</td>
<td>&lt;&lt;</td>
<td>sll</td>
</tr>
<tr>
<td>Shift right</td>
<td>&gt;&gt;</td>
<td>&gt;&gt;&gt;</td>
<td>srl</td>
</tr>
<tr>
<td>Bitwise AND</td>
<td>&amp;</td>
<td>&amp;</td>
<td>and, andi</td>
</tr>
<tr>
<td>Bitwise OR</td>
<td></td>
<td></td>
<td>or, ori</td>
</tr>
<tr>
<td>Bitwise NOT</td>
<td>~</td>
<td>~</td>
<td>nor</td>
</tr>
</tbody>
</table>

- Useful for extracting and inserting groups of bits in a word
Shift Operations

- **shamt**: how many positions to shift
- **Shift left logical**
  - Shift left and fill with 0 bits
  - $sll$ by $i$ bits multiplies by $2^i$
- **Shift right logical**
  - Shift right and fill with 0 bits
  - $srl$ by $i$ bits divides by $2^i$ (unsigned only)

<table>
<thead>
<tr>
<th>op</th>
<th>rs</th>
<th>rt</th>
<th>rd</th>
<th>shamt</th>
<th>funct</th>
</tr>
</thead>
<tbody>
<tr>
<td>6 bits</td>
<td>5 bits</td>
<td>5 bits</td>
<td>5 bits</td>
<td>5 bits</td>
<td>6 bits</td>
</tr>
</tbody>
</table>
AND Operations

- Useful to mask bits in a word
  - Select some bits, clear others to 0

```plaintext
and $t0, $t1, $t2
```

| $t2 | 0000 0000 0000 0000 0000 1101 1100 0000 |
| $t1 | 0000 0000 0000 0000 0011 1100 0000 0000 |
| $t0 | 0000 0000 0000 0000 0000 1100 0000 0000 |
OR Operations

- Useful to include bits in a word
  - Set some bits to 1, leave others unchanged

or \( t0, t1, t2 \)

| \( t2 \) | 0000 0000 0000 0000 0000 1101 1100 0000 |
| \( t1 \) | 0000 0000 0000 0000 0011 1100 0000 0000 |
| \( t0 \) | 0000 0000 0000 0000 0011 1101 1100 0000 |
NOT Operations

- Useful to invert bits in a word
  - Change 0 to 1, and 1 to 0
- MIPS has NOR 3-operand instruction
  - $a \text{ NOR } b = \text{ NOT } (a \text{ OR } b)$

`nor $t0, $t1, $zero`

Register 0: always read as zero

|$t1$| 0000 0000 0000 0000 0011 1100 0000 0000 |
|$t0$| 1111 1111 1111 1111 1100 0011 1111 1111 |
Conditional Operations

- **Branch to a labeled instruction if a condition is true**
  - Otherwise, continue sequentially
- **beq rs, rt, L1**
  - if (rs == rt) branch to instruction labeled L1;
- **bne rs, rt, L1**
  - if (rs != rt) branch to instruction labeled L1;
- **j L1**
  - unconditional jump to instruction labeled L1
Compiling If Statements

- **C code:**
  ```c
  if (i==j) f = g+h;
  else f = g-h;
  ```
  - $f, g, \ldots$ in $s0, s1, \ldots$

- **Compiled MIPS code:**
  ```assembly
  bne $s3, $s4, Else
  add $s0, $s1, $s2
  j Exit
  Else: sub $s0, $s1, $s2
  Exit: ...
  ```
  Assembler calculates addresses
Compiling Loop Statements

- C code:
  ```c
  while (save[i] == k) i += 1;
  ```
  - i in $s3, k in $s5, address of save in $s6

- Compiled MIPS code:
  ```mips
  Loop: sll $t1, $s3, 2
       add $t1, $t1, $s6
       lw $t0, 0($t1)
       bne $t0, $s5, Exit
       addi $s3, $s3, 1
       j Loop

  Exit: ...
  ```
Basic Blocks

- A basic block is a sequence of instructions with
  - No embedded branches (except at end)
  - No branch targets (except at beginning)

- A compiler identifies basic blocks for optimization
- An advanced processor can accelerate execution of basic blocks
More Conditional Operations

- Set result to 1 if a condition is true
  - Otherwise, set to 0

- `slt rd, rs, rt`
  - if (rs < rt) rd = 1; else rd = 0;

- `slti rt, rs, constant`
  - if (rs < constant) rt = 1; else rt = 0;

- Use in combination with `beq, bne`
  
  ```
  slt $t0, $s1, $s2  # if ($s1 < $s2)
  bne $t0, $zero, L  # branch to L
  ```
Branch Instruction Design

- Why not `blt`, `bge`, etc?
- Hardware for `<`, `≥`, … slower than `=`, `≠`
  - Combining with branch involves more work per instruction, requiring a slower clock
  - All instructions penalized!
- `beq` and `bne` are the common case
- This is a good design compromise
Signed vs. Unsigned

- Signed comparison: `slt`, `slti`
- Unsigned comparison: `sltu`, `sltui`

Example

- \$s0 = 1111 1111 1111 1111 1111 1111 1111 1111
- \$s1 = 0000 0000 0000 0000 0000 0000 0000 0001
- `slt $t0, $s0, $s1` # signed
  - \(-1 < +1 \Rightarrow t0 = 1\)
- `sltu $t0, $s0, $s1` # unsigned
  - \(+4,294,967,295 > +1 \Rightarrow t0 = 0\)
Procedure Calling

- Steps required
  1. Place parameters in registers
  2. Transfer control to procedure
  3. Acquire storage for procedure
  4. Perform procedure’s operations
  5. Place result in register for caller
  6. Return to place of call
Register Usage

- $a0 – $a3: arguments (reg’s 4 – 7)
- $v0, $v1: result values (reg’s 2 and 3)
- $t0 – $t9: temporaries
  - Can be overwritten by callee
- $s0 – $s7: saved
  - Must be saved/restored by callee
- $gp: global pointer for static data (reg 28)
- $sp: stack pointer (reg 29)
- $fp: frame pointer (reg 30)
- $ra: return address (reg 31)
Procedure Call Instructions

- Procedure call: jump and link
  \texttt{jal ProcedureLabel}
  - Address of following instruction put in $ra
  - Jumps to target address

- Procedure return: jump register
  \texttt{jr $ra}
  - Copies $ra to program counter
  - Can also be used for computed jumps
    - e.g., for case/switch statements
### Leaf Procedure Example

- **C code:**
  ```c
  int leaf_example (int g, h, i, j) {
    int f;
    f = (g + h) - (i + j);
    return f;
  }
  ```
  - Arguments g, ..., j in $a0, ..., $a3
  - f in $s0 (hence, need to save $s0 on stack)
  - Result in $v0
Leaf Procedure Example

- MIPS code:

  ```mips
  leaf_example:
      addi $sp, $sp, -4
      sw $s0, 0($sp)
      add $t0, $a0, $a1
      add $t1, $a2, $a3
      sub $s0, $t0, $t1
      add $v0, $s0, $zero
      lw $s0, 0($sp)
      addi $sp, $sp, 4
      jr $ra
  ```

  - Save $s0 on stack
  - Procedure body
  - Result
  - Restore $s0
  - Return
Non-Leaf Procedures

- Procedures that call other procedures
- For nested call, caller needs to save on the stack:
  - Its return address
  - Any arguments and temporaries needed after the call
- Restore from the stack after the call
Non-Leaf Procedure Example

- C code:
  ```c
  int fact (int n)
  {
    if (n < 1) return f;
    else return n * fact(n - 1);
  }
  ```
  - Argument n in $a0
  - Result in $v0
Non-Leaf Procedure Example

- **MIPS code:**

  ```mips
  fact:
  addi $sp, $sp, -8     # adjust stack for 2 items
  sw   $ra, 4($sp)      # save return address
  sw   $a0, 0($sp)      # save argument
  slti $t0, $a0, 1      # test for n < 1
  beq  $t0, $zero, L1
  addi $v0, $zero, 1    # if so, result is 1
  addi $sp, $sp, 8      # pop 2 items from stack
  jr   $ra              # and return
  L1:  addi $a0, $a0, -1     # else decrement n
  jal  fact             # recursive call
  lw   $a0, 0($sp)      # restore original n
  lw   $ra, 4($sp)      # and return address
  addi $sp, $sp, 8      # pop 2 items from stack
  mul  $v0, $a0, $v0    # multiply to get result
  jr   $ra              # and return
  ```
Local Data on the Stack

- Local data allocated by callee
  - e.g., C automatic variables
- Procedure frame (activation record)
  - Used by some compilers to manage stack storage
Memory Layout

- **Text**: program code
- **Static data**: global variables
  - e.g., static variables in C, constant arrays and strings
  - $gp$ initialized to address allowing ±offsets into this segment
- **Dynamic data**: heap
  - E.g., malloc in C, new in Java
- **Stack**: automatic storage
Character Data

- **Byte-encoded character sets**
  - ASCII: 128 characters
    - 95 graphic, 33 control
  - Latin-1: 256 characters
    - ASCII, +96 more graphic characters

- **Unicode: 32-bit character set**
  - Used in Java, C++ wide characters, ...
  - Most of the world’s alphabets, plus symbols
  - UTF-8, UTF-16: variable-length encodings
Byte/Halfword Operations

- Could use bitwise operations
- MIPS byte/halfword load/store
  - String processing is a common case
    \[
    \text{lb } rt, \text{ offset}(rs) \quad \text{lh } rt, \text{ offset}(rs)
    \]
    - Sign extend to 32 bits in rt
    \[
    \text{lbu } rt, \text{ offset}(rs) \quad \text{lhu } rt, \text{ offset}(rs)
    \]
    - Zero extend to 32 bits in rt
    \[
    \text{sb } rt, \text{ offset}(rs) \quad \text{sh } rt, \text{ offset}(rs)
    \]
    - Store just rightmost byte/halfword
String Copy Example

- C code (naïve):
  - Null-terminated string

```c
void strcpy (char x[], char y[])
{
    int i;
    i = 0;
    while (((x[i]=y[i])!='\0'))
        i += 1;
}
```
  - Addresses of x, y in $a0, $a1
  - i in $s0
String Copy Example

- **MIPS code:**

```mips
strcpy:
    addi $sp, $sp, -4      # adjust stack for 1 item
    sw   $s0, 0($sp)       # save $s0
    add  $s0, $zero, $zero # i = 0
L1:   add  $t1, $s0, $a1     # addr of y[i] in $t1
     lbu  $t2, 0($t1)       # $t2 = y[i]
     add  $t3, $s0, $a0     # addr of x[i] in $t3
     sb   $t2, 0($t3)       # x[i] = y[i]
    beq  $t2, $zero, L2    # exit loop if y[i] == 0
     addi $s0, $s0, 1       # i = i + 1
    j    L1                # next iteration of loop
L2:   lw   $s0, 0($sp)       # restore saved $s0
     addi $sp, $sp, 4       # pop 1 item from stack
    jr   $ra               # and return
```
32-bit Constants

- Most constants are small
  - 16-bit immediate is sufficient

- For the occasional 32-bit constant
  \( \text{lui \ rt, constant} \)
  - Copies 16-bit constant to left 16 bits of \( \text{rt} \)
  - Clears right 16 bits of \( \text{rt} \) to 0

\[
\text{lihi } \$s0, 61 \quad 0000 \ 0000 \ 0111 \ 1101 \ 0000 \ 0000 \ 0000 \ 0000 \\
\text{ori } \$s0, \$s0, 2304 \quad 0000 \ 0000 \ 0111 \ 1101 \ 0000 \ 1001 \ 0000 \ 0000
\]
Branch Addressing

- Branch instructions specify
  - Opcode, two registers, target address
- Most branch targets are near branch
  - Forward or backward

<table>
<thead>
<tr>
<th>op</th>
<th>rs</th>
<th>rt</th>
<th>constant or address</th>
</tr>
</thead>
<tbody>
<tr>
<td>6 bits</td>
<td>5 bits</td>
<td>5 bits</td>
<td>16 bits</td>
</tr>
</tbody>
</table>

- PC-relative addressing
  - Target address = PC + offset × 4
  - PC already incremented by 4 by this time
Jump Addressing

- Jump (j and jal) targets could be anywhere in text segment
  - Encode full address in instruction

<table>
<thead>
<tr>
<th>op</th>
<th>address</th>
</tr>
</thead>
<tbody>
<tr>
<td>6 bits</td>
<td>26 bits</td>
</tr>
</tbody>
</table>

- (Pseudo)Direct jump addressing
  - Target address = PC_{31...28} : (address \times 4)
Target Addressing Example

- Loop code from earlier example
  - Assume Loop at location 80000

<table>
<thead>
<tr>
<th>Instruction</th>
<th>Address</th>
<th>Symbol</th>
<th>Symbol</th>
<th>Symbol</th>
<th>Symbol</th>
<th>Symbol</th>
</tr>
</thead>
<tbody>
<tr>
<td>sll $t1, $s3, 2</td>
<td>80000</td>
<td>t1</td>
<td>s3</td>
<td>2</td>
<td></td>
<td></td>
</tr>
<tr>
<td>add $t1, $t1, $s6</td>
<td>80004</td>
<td>0</td>
<td>9</td>
<td>22</td>
<td>9</td>
<td>0</td>
</tr>
<tr>
<td>lw $t0, 0($t1)</td>
<td>80008</td>
<td>35</td>
<td>9</td>
<td>8</td>
<td>0</td>
<td></td>
</tr>
<tr>
<td>bne $t0, $s5, Exit</td>
<td>80012</td>
<td>5</td>
<td>8</td>
<td>21</td>
<td>2</td>
<td></td>
</tr>
<tr>
<td>addi $s3, $s3, 1</td>
<td>80016</td>
<td>8</td>
<td>19</td>
<td>19</td>
<td>1</td>
<td></td>
</tr>
<tr>
<td>j Loop</td>
<td>80020</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td>2</td>
</tr>
<tr>
<td>Exit: ...</td>
<td>80024</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td>20000</td>
</tr>
</tbody>
</table>
Branching Far Away

- If branch target is too far to encode with 16-bit offset, assembler rewrites the code

- Example

  ```
  beq $s0,$s1, L1
  ↓
  bne $s0,$s1, L2
  j L1
  L2: ...
  ```
Addressing Mode Summary

1. Immediate addressing
   \[ \text{op} \quad \text{rs} \quad \text{rt} \quad \text{Immediate} \]

2. Register addressing
   \[ \text{op} \quad \text{rs} \quad \text{rt} \quad \text{rd} \quad \ldots \quad \text{funct} \]
   Registers
   \[ \text{Register} \]

3. Base addressing
   \[ \text{op} \quad \text{rs} \quad \text{rt} \quad \text{Address} \]
   \[ \text{Register} \quad + \]
   Memory
   \[ \text{Byte} \quad \text{Halfword} \quad \text{Word} \]

4. PC-relative addressing
   \[ \text{op} \quad \text{rs} \quad \text{rt} \quad \text{Address} \]
   \[ \text{PC} \quad + \]
   Memory
   \[ \text{Word} \]

5. Pseudodirect addressing
   \[ \text{op} \quad \text{Address} \]
   \[ \text{PC} \]
   Memory
   \[ \text{Word} \]
Synchronization

- Two processors sharing an area of memory
  - P1 writes, then P2 reads
  - Data race if P1 and P2 don’t synchronize
    - Result depends of order of accesses

- Hardware support required
  - Atomic read/write memory operation
  - No other access to the location allowed between the read and write

- Could be a single instruction
  - E.g., atomic swap of register ↔ memory
  - Or an atomic pair of instructions
Synchronization in MIPS

- **Load linked:** `ll rt, offset(rs)`
- **Store conditional:** `sc rt, offset(rs)`
  - Succeeds if location not changed since the `ll`
    - Returns 1 in `rt`
  - Fails if location is changed
    - Returns 0 in `rt`

- **Example: atomic swap**
  (to test/set lock variable)
  ```
  try: add $t0,$zero,$s4 ; copy exchange value
     ll $t1,0($s1) ; load linked
     sc $t0,0($s1) ; store conditional
     beq $t0,$zero,try ; branch store fails
    add $s4,$zero,$t1 ; put load value in $s4
  ```
Many compilers produce object modules directly.

$\begin{align*}
\text{C program} & \quad \xrightarrow{\text{Compiler}} \quad \text{Assembly language program} \\
\text{Assembler} & \quad \xrightarrow{\text{Object: Machine language module}} \quad \text{Object: Library routine (machine language)} \\
\text{Linker} & \quad \xrightarrow{\text{Executable: Machine language program}} \\
\text{Loader} & \quad \xrightarrow{\text{Memory}} \\
\end{align*}$
Assembler Pseudoinstructions

- Most assembler instructions represent machine instructions one-to-one
- Pseudoinstructions: figments of the assembler’s imagination

\[
\begin{align*}
\text{move } &\, \$t0, \, \$t1 \quad \rightarrow \quad \text{add } \$t0, \, \$\text{zero}, \, \$t1 \\
\text{blt } &\, \$t0, \, \$t1, \, L \quad \rightarrow \quad \text{slt } \$at, \, \$t0, \, \$t1 \\
&\quad \quad \text{bne } \$at, \, \$\text{zero}, \, L
\end{align*}
\]

- \$at (register 1): assembler temporary
Producing an Object Module

- Assembler (or compiler) translates program into machine instructions
- Provides information for building a complete program from the pieces
  - Header: described contents of object module
  - Text segment: translated instructions
  - Static data segment: data allocated for the life of the program
  - Relocation info: for contents that depend on absolute location of loaded program
  - Symbol table: global definitions and external refs
  - Debug info: for associating with source code
Linking Object Modules

- Produces an executable image
  1. Merges segments
  2. Resolve labels (determine their addresses)
  3. Patch location-dependent and external refs

- Could leave location dependencies for fixing by a relocating loader
  - But with virtual memory, no need to do this
  - Program can be loaded into absolute location in virtual memory space
Loading a Program

- Load from image file on disk into memory
  1. Read header to determine segment sizes
  2. Create virtual address space
  3. Copy text and initialized data into memory
     - Or set page table entries so they can be faulted in
  4. Set up arguments on stack
  5. Initialize registers (including $sp, $fp, $gp)
  6. Jump to startup routine
     - Copies arguments to $a0, … and calls main
     - When main returns, do exit syscall
Dynamic Linking

- Only link/load library procedure when it is called
  - Requires procedure code to be relocatable
  - Avoids image bloat caused by static linking of all (transitively) referenced libraries
  - Automatically picks up new library versions
Lazy Linkage

Indirection table

Stub: Loads routine ID, Jump to linker/loader

Linker/loader code

Dynamically mapped code

a. First call to DLL routine

b. Subsequent calls to DLL routine
Starting Java Applications

- Simple portable instruction set for the JVM
- Interprets bytecodes
- Compiles bytecodes of “hot” methods into native code for host machine
- Just In Time compiler
- Java Virtual Machine
- Class files (Java bytecodes)
- Java library routines (machine language)
- Compiled Java methods (machine language)
C Sort Example

- Illustrates use of assembly instructions for a C bubble sort function
- Swap procedure (leaf)
  
  ```c
  void swap(int v[], int k)
  {
      int temp;
      temp = v[k];
      v[k] = v[k+1];
      v[k+1] = temp;
  }
  
  • v in $a0, k in $a1, temp in $t0
  ```
The Procedure Swap

\[
\begin{align*}
\text{swap:} & \quad \text{sll } \text{t1}, \text{a1, 2} \quad & \# \text{t1} = k \ast 4 \\
& \quad \text{add } \text{t1, a0, t1} \quad & \# \text{t1} = v + (k \ast 4) \\
& \quad & \# \text{(address of v}[k]\text{])} \\
& \quad \text{lw } \text{t0, 0(t1)} \quad & \# \text{t0 (temp) = v}[k]\text{] } \\
& \quad \text{lw } \text{t2, 4(t1)} \quad & \# \text{t2 = v}[k+1]\text{] } \\
& \quad \text{sw } \text{t2, 0(t1)} \quad & \# \text{v}[k] = \text{t2 (v}[k+1]\text{]} \\
& \quad \text{sw } \text{t0, 4(t1)} \quad & \# \text{v}[k+1] = \text{t0 (temp)} \\
& \quad \text{jr } \text{ra} \quad & \# \text{return to calling routine}
\end{align*}
\]
The Sort Procedure in C

- Non-leaf (calls swap)
  ```c
  void sort (int v[], int n)
  {
    int i, j;
    for (i = 0; i < n; i += 1) {
      for (j = i - 1;
          j >= 0 && v[j] > v[j + 1];
          j -= 1) {
        swap(v,j);
      }
    }
  }
  ```

  • $v$ in $a0$, $k$ in $a1$, $i$ in $s0$, $j$ in $s1$
The Procedure Body

move $s2, $a0           # save $a0 into $s2
move $s3, $a1           # save $a1 into $s3
move $s0, $zero         # i = 0

for1tst: slt  $t0, $s0, $s3      # $t0 = 0 if $s0 ≥ $s3 (i ≥ n)
    beq  $t0, $zero, exit1  # go to exit1 if $s0 ≥ $s3 (i ≥ n)
    addi $s1, $s0, –1       # j = i – 1

for2tst: slti $t0, $s1, 0        # $t0 = 1 if $s1 < 0 (j < 0)
    bne  $t0, $zero, exit2  # go to exit2 if $s1 < 0 (j < 0)
    sll  $t1, $s1, 2        # $t1 = j * 4
    add  $t2, $s2, $t1      # $t2 = v + (j * 4)
    lw   $t3, 0($t2)        # $t3 = v[j]
    lw   $t4, 4($t2)        # $t4 = v[j + 1]
    slt  $t0, $t4, $t3      # $t0 = 0 if $t4 ≥ $t3
    beq  $t0, $zero, exit2  # go to exit2 if $t4 ≥ $t3

move $a0, $s2           # 1st param of swap is v (old $a0)
move $a1, $s1           # 2nd param of swap is j
jal  swap               # call swap procedure
    addi $s1, $s1, –1       # j -= 1
    j    for2tst            # jump to test of inner loop

exit2:   addi $s0, $s0, 1        # i += 1
    j    for1tst            # jump to test of outer loop
sort:    addi $sp, $sp, -20  # make room on stack for 5 registers
sw $ra, 16($sp)            # save $ra on stack
sw $s3, 12($sp)           # save $s3 on stack
sw $s2, 8($sp)            # save $s2 on stack
sw $s1, 4($sp)            # save $s1 on stack
sw $s0, 0($sp)            # save $s0 on stack
...                      # procedure body
...                      
exit1: lw $s0, 0($sp)      # restore $s0 from stack
lw $s1, 4($sp)            # restore $s1 from stack
lw $s2, 8($sp)            # restore $s2 from stack
lw $s3, 12($sp)           # restore $s3 from stack
lw $ra, 16($sp)           # restore $ra from stack
addi $sp, $sp, 20         # restore stack pointer
jr $ra                    # return to calling routine
Effect of Compiler Optimization

Compiled with gcc for Pentium 4 under Linux
Effect of Language and Algorithm

**Bubblesort Relative Performance**

<table>
<thead>
<tr>
<th></th>
<th>C/none</th>
<th>C/O1</th>
<th>C/O2</th>
<th>C/O3</th>
<th>Java/int</th>
<th>Java/JIT</th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td>0.5</td>
<td>2.5</td>
<td>2.5</td>
<td>2.5</td>
<td>0.5</td>
<td>2.5</td>
</tr>
</tbody>
</table>

**Quicksort Relative Performance**

<table>
<thead>
<tr>
<th></th>
<th>C/none</th>
<th>C/O1</th>
<th>C/O2</th>
<th>C/O3</th>
<th>Java/int</th>
<th>Java/JIT</th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td>1.0</td>
<td>2.0</td>
<td>2.0</td>
<td>2.0</td>
<td>0.0</td>
<td>0.0</td>
</tr>
</tbody>
</table>

**Quicksort vs. Bubblesort Speedup**

<table>
<thead>
<tr>
<th></th>
<th>C/none</th>
<th>C/O1</th>
<th>C/O2</th>
<th>C/O3</th>
<th>Java/int</th>
<th>Java/JIT</th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td>3000</td>
<td>2500</td>
<td>2000</td>
<td>1500</td>
<td>1000</td>
<td>500</td>
</tr>
</tbody>
</table>
Lessons Learnt

- Instruction count and CPI are not good performance indicators in isolation
- Compiler optimizations are sensitive to the algorithm
- Java/JIT compiled code is significantly faster than JVM interpreted
  - Comparable to optimized C in some cases
- Nothing can fix a dumb algorithm!
Arrays vs. Pointers

- Array indexing involves
  - Multiplying index by element size
  - Adding to array base address

- Pointers correspond directly to memory addresses
  - Can avoid indexing complexity
Example: Clearing and Array

Example: Clearing and Array

```c
Example: Clearing and Array
clear1(int array[], int size) {
    int i;
    for (i = 0; i < size; i += 1)
        array[i] = 0;
}
clear2(int *array, int size) {
    int *p;
    for (p = &array[0]; p < &array[size];
        p = p + 1)
        *p = 0;
}
```

```assembly
Example: Clearing and Array
move $t0,$zero   # i = 0
loop1: sll $t1,$t0,2    # $t1 = i * 4
        add $t2,$a0,$t1  # $t2 =
                        #   &array[i]
        sw $zero, 0($t2) # array[i] = 0
        addi $t0,$t0,1   # i = i + 1
        slt $t3,$t0,$a1  # $t3 =
                        #   (i < size)
        bne $t3,$zero,loop1 # if (…)
                        #   goto loop1
move $t0,$a0      # p = & array[0]
sll $t1,$a1,2     # $t1 = size * 4
        add $t2,$a0,$t1  # $t2 =
                        #   &array[size]
loop2: sw $zero,0($t0) # Memory[p] = 0
        addi $t0,$t0,4   # p = p + 4
        slt $t3,$t0,$t2  # $t3 =
                        #   (p<&array[size])
        bne $t3,$zero,loop2 # if (…)
                        #   goto loop2
```
Comparison of Array vs. Ptr

- Multiply “strength reduced” to shift
- Array version requires shift to be inside loop
  - Part of index calculation for incremented i
  - c.f. incrementing pointer
- Compiler can achieve same effect as manual use of pointers
  - Induction variable elimination
  - Better to make program clearer and safer
### ARM & MIPS Similarities

- ARM: the most popular embedded core
- Similar basic set of instructions to MIPS

<table>
<thead>
<tr>
<th></th>
<th>ARM</th>
<th>MIPS</th>
</tr>
</thead>
<tbody>
<tr>
<td>Date announced</td>
<td>1985</td>
<td>1985</td>
</tr>
<tr>
<td>Instruction size</td>
<td>32 bits</td>
<td>32 bits</td>
</tr>
<tr>
<td>Address space</td>
<td>32-bit flat</td>
<td>32-bit flat</td>
</tr>
<tr>
<td>Data alignment</td>
<td>Aligned</td>
<td>Aligned</td>
</tr>
<tr>
<td>Data addressing modes</td>
<td>9</td>
<td>3</td>
</tr>
<tr>
<td>Registers</td>
<td>15 (\times) 32-bit</td>
<td>31 (\times) 32-bit</td>
</tr>
<tr>
<td>Input/output</td>
<td>Memory mapped</td>
<td>Memory mapped</td>
</tr>
</tbody>
</table>
Compare and Branch in ARM

- Uses condition codes for result of an arithmetic/logical instruction
  - Negative, zero, carry, overflow
  - Compare instructions to set condition codes without keeping the result

- Each instruction can be conditional
  - Top 4 bits of instruction word: condition value
  - Can avoid branches over single instructions
Instruction Encoding

**Register-register**

- **ARM**:
  - 31 28 27 20 19 16 15 12 11 4 3 0
  - Op3 Op2 Rs1 Op4 Rd Rs2 Op5 Rs2

- **MIPS**: 31 26 25 21 20 16 15 11 10 6 5 0
  - Op3 Rs1 Rs2 Op5 Rd Const3 Op5

**Data transfer**

- **ARM**: 31 28 27 20 19 16 15 12 11 0
  - Op3 Op2 Rs1 Op4 Rd Const12

- **MIPS**: 31 26 25 21 20 16 15 0
  - Op3 Rs1 Rd2 Const18

**Branch**

- **ARM**: 31 28 27 24 23 0
  - Op3 Op4 Const24

- **MIPS**: 31 26 25 21 20 16 15 0
  - Op3 Rs1 Op3 Rs2 Const18

**Jump/Call**

- **ARM**: 31 28 27 24 23 0
  - Op3 Op4 Const24

- **MIPS**: 31 26 25 21 20 16 15 0
  - Op3 Const26

Legend:
- Opcode
- Register
- Constant
The Intel x86 ISA

- Evolution with backward compatibility
  - 8080 (1974): 8-bit microprocessor
    - Accumulator, plus 3 index-register pairs
  - 8086 (1978): 16-bit extension to 8080
    - Complex instruction set (CISC)
  - 8087 (1980): floating-point coprocessor
    - Adds FP instructions and register stack
  - 80286 (1982): 24-bit addresses, MMU
    - Segmented memory mapping and protection
  - 80386 (1985): 32-bit extension (now IA-32)
    - Additional addressing modes and operations
    - Paged memory mapping as well as segments
The Intel x86 ISA

Further evolution…

- i486 (1989): pipelined, on-chip caches and FPU
  - Compatible competitors: AMD, Cyrix, …
- Pentium (1993): superscalar, 64-bit datapath
  - Later versions added MMX (Multi-Media eXtension) instructions
  - The infamous FDIV bug
  - New microarchitecture (see Colwell, *The Pentium Chronicles*)
- Pentium III (1999)
  - Added SSE (Streaming SIMD Extensions) and associated registers
- Pentium 4 (2001)
  - New microarchitecture
  - Added SSE2 instructions
The Intel x86 ISA

And further...

- **AMD64 (2003):** extended architecture to 64 bits
- **EM64T** – Extended Memory 64 Technology (2004)
  - AMD64 adopted by Intel (with refinements)
  - Added SSE3 instructions
- **Intel Core (2006)**
  - Added SSE4 instructions, virtual machine support
- **AMD64 (announced 2007):** SSE5 instructions
  - Intel declined to follow, instead...
- **Advanced Vector Extension (announced 2008)**
  - Longer SSE registers, more instructions

If Intel didn’t extend with compatibility, its competitors would!

- Technical elegance ≠ market success
# Basic x86 Registers

<table>
<thead>
<tr>
<th>Name</th>
<th>Use</th>
</tr>
</thead>
<tbody>
<tr>
<td>EAX</td>
<td>GPR 0</td>
</tr>
<tr>
<td>ECX</td>
<td>GPR 1</td>
</tr>
<tr>
<td>EDX</td>
<td>GPR 2</td>
</tr>
<tr>
<td>EBX</td>
<td>GPR 3</td>
</tr>
<tr>
<td>ESP</td>
<td>GPR 4</td>
</tr>
<tr>
<td>EBP</td>
<td>GPR 5</td>
</tr>
<tr>
<td>ESI</td>
<td>GPR 6</td>
</tr>
<tr>
<td>EDI</td>
<td>GPR 7</td>
</tr>
<tr>
<td>CS</td>
<td>Code segment pointer</td>
</tr>
<tr>
<td>SS</td>
<td>Stack segment pointer (top of stack)</td>
</tr>
<tr>
<td>DS</td>
<td>Data segment pointer 0</td>
</tr>
<tr>
<td>ES</td>
<td>Data segment pointer 1</td>
</tr>
<tr>
<td>FS</td>
<td>Data segment pointer 2</td>
</tr>
<tr>
<td>GS</td>
<td>Data segment pointer 3</td>
</tr>
<tr>
<td>EIP</td>
<td>Instruction pointer (PC)</td>
</tr>
<tr>
<td>EFLAGS</td>
<td>Condition codes</td>
</tr>
</tbody>
</table>
### Basic x86 Addressing Modes

- **Two operands per instruction**

<table>
<thead>
<tr>
<th>Source/dest operand</th>
<th>Second source operand</th>
</tr>
</thead>
<tbody>
<tr>
<td>Register</td>
<td>Register</td>
</tr>
<tr>
<td>Register</td>
<td>Immediate</td>
</tr>
<tr>
<td>Register</td>
<td>Memory</td>
</tr>
<tr>
<td>Memory</td>
<td>Register</td>
</tr>
<tr>
<td>Memory</td>
<td>Immediate</td>
</tr>
</tbody>
</table>

- **Memory addressing modes**
  - Address in register
    - Address = \( R_{\text{base}} + \text{displacement} \)
    - Address = \( R_{\text{base}} + 2^{\text{scale}} \times R_{\text{index}} \) (scale = 0, 1, 2, or 3)
    - Address = \( R_{\text{base}} + 2^{\text{scale}} \times R_{\text{index}} + \text{displacement} \)
### Variable length encoding

- Postfix bytes specify addressing mode
- Prefix bytes modify operation
  - Operand length, repetition, locking, …

<table>
<thead>
<tr>
<th>Instruction</th>
<th>Format</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>a. JE EIP + displacement</td>
<td>4 4 8</td>
<td>JE Condition Displacement</td>
</tr>
<tr>
<td>b. CALL</td>
<td>8 32</td>
<td>CALL Offset</td>
</tr>
<tr>
<td>c. MOV EBX, [EDI + 45]</td>
<td>6 1 1 8 8</td>
<td>MOV d w r/m Postbyte Displacement</td>
</tr>
<tr>
<td>d. PUSH ESI</td>
<td>5 3</td>
<td>PUSH Reg</td>
</tr>
<tr>
<td>e. ADD EAX, #6765</td>
<td>4 3 1 32</td>
<td>ADD Reg w Immediate</td>
</tr>
<tr>
<td>f. TEST EDX, #42</td>
<td>7 1 8 32</td>
<td>TEST w Postbyte Immediate</td>
</tr>
</tbody>
</table>
Implementing IA-32

- Complex instruction set makes implementation difficult
  - Hardware translates instructions to simpler microoperations
    - Simple instructions: 1–1
    - Complex instructions: 1–many
  - Microengine similar to RISC
  - Market share makes this economically viable

- Comparable performance to RISC
  - Compilers avoid complex instructions
Fallacies

- **Powerful instruction ⇒ higher performance**
  - Fewer instructions required
  - But complex instructions are hard to implement
    - May slow down all instructions, including simple ones
  - Compilers are good at making fast code from simple instructions

- **Use assembly code for high performance**
  - But modern compilers are better at dealing with modern processors
  - More lines of code ⇒ more errors and less productivity
Fallacies

- Backward compatibility $\Rightarrow$ instruction set doesn’t change
  - But they do accrete more instructions
Pitfalls

- Sequential words are not at sequential addresses
  - Increment by 4, not by 1!

- Keeping a pointer to an automatic variable after procedure returns
  - e.g., passing pointer back via an argument
  - Pointer becomes invalid when stack popped
Concluding Remarks

- Design principles
  1. Simplicity favors regularity
  2. Smaller is faster
  3. Make the common case fast
  4. Good design demands good compromises

- Layers of software/hardware
  - Compiler, assembler, hardware

- MIPS: typical of RISC ISAs
  - c.f. x86
Concluding Remarks

- Measure MIPS instruction executions in benchmark programs
  - Consider making the common case fast
  - Consider compromises

<table>
<thead>
<tr>
<th>Instruction class</th>
<th>MIPS examples</th>
<th>SPEC2006 Int</th>
<th>SPEC2006 FP</th>
</tr>
</thead>
<tbody>
<tr>
<td>Arithmetic</td>
<td>add, sub, addi</td>
<td>16%</td>
<td>48%</td>
</tr>
<tr>
<td>Data transfer</td>
<td>lw, sw, lb, lbu, lh, lhu, sb, lui</td>
<td>35%</td>
<td>36%</td>
</tr>
<tr>
<td>Logical</td>
<td>and, or, nor, andi, ori, sll, srl</td>
<td>12%</td>
<td>4%</td>
</tr>
<tr>
<td>Cond. Branch</td>
<td>beq, bne, slt, slti, sltiu</td>
<td>34%</td>
<td>8%</td>
</tr>
<tr>
<td>Jump</td>
<td>j, jr, jal</td>
<td>2%</td>
<td>0%</td>
</tr>
</tbody>
</table>